Tutorial 26: complément pour les strings

Afficher les strings avec l'ASM

Introduction

Tout ce qui se passe avec les strings est plus facile avec la table des symboles ainsi que d'autres variables, mais toutes les manières sont bonnes. Ce tutorial va afficher les strings de 2 façons - en utilisant la mémoire cache (buffer) ou en ne l'utilisant pas.

Buffer?

Le "buffer", la mémoire cache, c'est une partie de la mémoire qui vous permet de garder des données pendant un moment. Un peu comme le copier/coller.

Nouvelles commandes

La seule nouvelle commande est:

_ldhlind

- h = (hl), l = (hl+1).

Le code

Ce code utilise le buffer

.NOLIST

#define equ .equ

#define EQU .equ

#define end .end

#include "ti83asm.inc"

#include "tokens.inc"

.LIST

#define buffer   8265h

.org 9327h

     call _zerooop1    ; load var name

     ld hl,strngname   ;

     ld de,op1         ;

     ld bc,3           ;

     ldir              ;

     call _chkfindsym  ; look it up

     ret c             ; return if it is not found

     ex de,hl          ; exchange de and hl

     ld c,(hl)         ; get LSB of the size

     inc hl            ; next byte

     ld b,(hl)         ; get MSB of the size

     inc hl            ; next byte

     ld de,buffer      ; buffer

     ldir              ; copy data

     ld a,0            ;

     ld (de),a         ; add 0 terminator

     ld hl,buffer      ; address to buffer

     call _puts        ; display string

     call _newline     ; move to a new line

     ret

strngname:

     .db $04,$AA,$00  ; object type strng, var type strng, var Str1

.end

END

Ce code n'utilse pas le buffer pour afficher le string

.NOLIST

#define equ .equ

#define EQU .equ

#define end .end

#include "ti83asm.inc"

#include "tokens.inc"

.LIST

_errundefined .equ 467Bh

strngobj .equ 04h

.org 9327h

     call _zerooop1      ; clear op1

     ld hl,op1           ; load var name for Str1 into op1

     ld (hl),strngobj    ;

     inc hl              ;

     ld (hl),tvarstrng   ;

     inc hl              ;

     ld (hl),tstr1       ;

     call _chkfindsym    ; look it up

     jp c,_errundefined  ; error if not found

     push de             ;

     pop hl              ; hl -> string address

     call _ldhlind       ; hl -> string size

     push hl             ;

     pop bc              ; bc -> string size

     inc de              ;

     inc de              ; de -> string data

sloop:

     ld a,(de)           ; get character

     call _putc          ; draw character

     inc de              ; next character

     dec bc              ; decrement counter

     ld a,b              ; see if bc is at 0

     or c                ;

     jr nz,sloop         ;

     call _newline       ; move to a new line

     ret

.end

END

Ian a écrit ces 2 programmes… merci à lui. Une chose qui doit être expliquée c'est le LSB et le MSB

LSB, MSB

LSB: "Less Significant Bbyte" C'est le bit le moins significatif

MSB: "Most Significant Byte" C'est le bit le plus significatif

exemples:    10101

1999

                     |      |

|      |  

               MSB    LSB              MSB    LSB

Conclusion

Ce programme est utile en combinant l'autre programme sur l'écriture des highscores pour les jeux etc… Str1 est souvent utilisé, prenez en un autre pour les scores de vos jeux.

Merci à Ian Graf

