Tutorial 50 : Programmes d’interruption

Programme en arrière plan

Introduction

50 tutorials ! Wow je ne pensais pas que j’en écrirai autant. Bien ça se fête ( !) Je vais faire un tutorial long et compliqué - désolé, c’est le seul moyen pour que ça marche. Ce tutorial va vous apprendre comment faire des interruptions de programmes. Les programmes d’interruptions sont des programmes qui s ‘exécutent en arrière plan, ils sont appelés tous les 200ème de seconde et l’OS de TI s’exécute normalement en tâche de fond.

Concepts

Le processeur Z80 a plusieurs modes d’interruption - 0, 1, 2. Ils sont spécifiés avec IMx. Le mode 0 n’est pas utilisé sur la calculatrice, le mode 1 est utilisé la plupart du temps. Le processeur regarde toujours à l’adresse $0038 pour l’interruption maître. Encore une fois nous n’utilisons pas ceci. C’est IM2 qui nous intéresse. Le processeur utilise un vecteur sur une adresse mémoire pour spécifier l’interruption maître. Le MSB de l’adresse de la table des vecteurs est stocké dans le registre I (Note : Cela n’a aucun rapport avec les registres IX et IY). Le problème c’est que le système d’interruption donne le LSB, donc quasiment un nombre aléatoire. Afin de toujours aller à la bonne adresse mémoire nous devons avoir 257 octets de mémoire (de $xx00 à $xxFF, où xx est la valeur de I) contenant le même octet $XY.

Pourquoi ? Pensez à ceci, l’interruption a lieu tout les 200ème de seconde. Le MSB est tout ce qu’il y a dans I (disons $82 par exemple). Le LSB est aléatoire - disons $32. La calculatrice regarde à $8232 et trouve $XYXY, parce qu’on a remplit $8200 à $82FF avec cet octet. Donc la calculatrice l’appelle $XYXY ! Qu’est-ce qui se passe si le LSB = $21 ? Bien à $8221 ce serait toujours $XYXY, donc la calculatrice l’appellerai toujours. Amusant ! :) ...

Quand vous programmez un programme d’interruption vous devez garder en tête quelques trucs. Vous devez toujours préserver les registres ! ! maintenant comment sur terre pouvez vous faire un programme sans utiliser un simple registre ? Vous ne pouvez pas... ce que vous faites c’est de sauvegarder tous les registres avec les ombres des registres. Le sombres des registres sont ce que leur nom implique - des copies des registres. Bien vous devez trouver une place appropriée dans la RAM de la TI pour stocker votre programme. C’est une tâche quelque peu difficile mais nous l’avons déjà fait dans quelques programmes, nous pouvons donc regarder leur code.

APD

Le programme auquel nous allons regarder est Zcapture de Ian Graf. Il capture l’écran et le met dans Pic1. Ce qui est bien avec lui c’est qu’il supporte l’APD (automatique powerdown) extinction automatique. Quittez votre TI pendant un moment et il s’éteint automatiquement. Si vous exécutez un programme d’interruption, APD sera désactivé tant que vous ne le réactiverai pas vous même. Le programme de Ian nous montre comment faire ça. Allez place au ocde !...

Nouvelles commandes


IXx
; change le mode d’interruption x


EXX
; intervertit BC, DE, HL avec les copies de registres

Le programmes suivant est probablement le plus complexe que nous avons vu, s’il vous plaît accompagner moi. Avec les commentaires de Ian et les miens que j’ai intercalé, vous devriez comprendre.

;-------------= Capture 83 =------------

; Author:       Ian Graf

;               (ian_graf@geocities.com)

; Version:      1.0

; Platform:     TI-OS

; Date:         28/8/98

;---------------------------------------

        .nolist

        #include        "capture.inc"

Je reviendrai sur ce fichier à la fin, il définit toutes les commandes que Ian utilise

#define         lcdbusy         delay-interrupt_start+$8484

.list

.org    usermem

 im      1                       ; make sure it isn't running in im 2
On retourne juste IM1 au cas où un programme le modifie. Cela peut avoir de mauvais effets, crash de la TI, erreur de mémoire etc... choses que nous n’aimons pas.

        ld      hl,capture              ; display "Capture 83"

        call    puts                    ;

        call    newline                 ;

        ld      hl,by                   ; display "By Ian Graf"

        call    puts                    ;

        call    newline                 ;

        ld      a,(state)               ; check if it is running
Vous vous souvenez du tutorial avec l’écriture dans un programme ? Ian utilise la même technique. Il écrit dans le programme s’il s’exécute ou non. Cela nous montre comment basculer le programme de on à off en ‘ l’exécutant ’.

or      a                       ;

        jr      nz,disable_interrupt    ;

;---------= enable interrupt =----------

        enable_interrupt:

        ld      hl,enabled              ; dipslay "Enabled"

        call    puts                    ;

        call    newline                 ;

        call    zeroop1                 ;

        ld      hl,op1                  ;

        ld      (hl),pictobj            ; pic1

        inc     hl                      ;

        ld      (hl),tvarpict           ;

        inc     hl                      ;

        ld      (hl),tpic1              ;

        call    chkfindsym              ; look it up

        call    c,createpict            ; create it if not found

Comme le programme nécessite une image, Ian vérifie si elle existe, sinon il la crée. Ceci a déjà été vu et vous devriez le connaître.

        ld      hl,$8300                ; set up interrupt vector

        ld      de,$8301                ;

        ld      (hl),$84                ;

        ld      bc,$0100                ;

        ldir                   

Ok Ian stocke $8300 dans hl et l’octet après hl dans de. Après il met $84 dans hl , 256 dans bc et utilise LDIR pour charger les 256 octets (257 en tout) suivant avec $84.

        ld      hl,interrupt_start      ; copy interrupt program

        ld      de,$8484                ;

        ld      bc,interrupt_end-interrupt_start  ;

        ldir     

Charge l'adresse de interrupt_start (c'est un label un peu plus bas) à l'adresse $8484. BC a besoin d'être le nombre d'octets, donc simplement soustraire l'adresse à la fin du label avec celle du début

xor     a                       ; [LN] is not pressed

       
 ld      (keydown),a             ;

       
 ld      a,$83                   ; point i to interrupt vector

       
 ld      i,a                     ;

Notre table des vecteurs commence à $83 et le stocke dans i.

im      2                       ;

Change le mode d'interruption

ld      a,1                     ; set state to enabled

        
jr      store_state             ;

;---------= disable interrupt =---------

disable_interrupt:

        
ld      hl,disabled             ; display "Disabled"

        
call    puts                    ;

        
call    newline                 ;

        
xor     a                       ; set state to disabled

;------------= store state =------------

; input:        a - state (boolean)

; output:       state written back

store_state:

        push    af                      ; store state

        call    zeroop1                 ; clear op1

        ld      de,op1                  ; copy name

        ld      hl,prog                 ;

        ld      bc,9                    ;

        ldir                            ;

        call    chkfindsym              ; look up program

        inc     de                      ;

        inc     de                      ;

        ld      hl,state-$9327          ; get offset to state

        add     hl,de                   ;

        pop     af                      ; restore state

        ld      (hl),a                  ; write state

        ret                             ;

Ok vous devez comprendre tout ceci avec le tutorial d'écriture dans un programme. Maintenant notez comment nous n'appelons jamais directement le programme d'interruption. La partie exécutive de ce programme est plus un chargement qui copie tout dans la mémoire puis se termine, en laissant tout à sa place. Tout le code compliqué est maintenant fait, place au programme d'interruption.

;---------= interrupt program =---------

interrupt_start:

        di                              ; disable interrupts

        ex      af,af'                  ;

        exx                             ;

Change tout les registres.

ld      a,($800A)               ; check for apd


        cp      $11                     ;

        jr      c,interrupt_off         ;

Si la TI doit s'éteindre, le faire !!

        ld      a,%11111111             ; reset key port

        out     (1),a                   ;

        ld      a,%11011111             ; check for [LN] press

Active le Group6, voir le tutorial sur les entrées directes.

out     (1),a                   ;

        
in      a,(1)                   ;

        
bit     2,a                     ;

        
jr      z,key                   ;

Oh Ian n'est pas malin. L'équation de la touche LN est 251 qui est 11111011. Donc il vérifie si le deuxième bit est à 0 ou non. (souvenons nous que les bits commençants à 0… et qu'il n'y a pas de confusion possible avec les autres touches puisqu'on se limite à un groupe de touches)

xor     a                       ; keep [LN] from repeating

        
ld      (keydown),a             ;

        
jr      exit_interrupt          ;

Si aucune touche n'est précisée, fin. le programme est exécuter 200 fois par seconde

key:    ld      a,(keydown)             ;

        or      a                       ;

        jr      nz,exit_interrupt       ;

        ld      a,1                     ;

        ld      (keydown),a             ;

        ld      hl,symtable             ; search for pic in vat

picl:   ld      a,(hl)                  ; get type

        and     %00011111               ; mask system bits

        cp      pictobj                 ; pic data

        jr      z,pic                   ;

        ld      de,-6                   ; next entry

Souvenez-vous que la table des symboles fonctionne en inverse, d'où le -6. vous devriez comprendre la plupart des choses.

add     hl,de                   ;

        jr      picl                    ;

pic:    dec     hl                      ; get data pointer

        ld      e,(hl)                  ;

        dec     hl                      ;

        ld      d,(hl)                  ;

        inc     de                      ; skip size

        inc     de                      ;

        ex      de,hl                   ; hl points to pic data

Tous les pointeurs de l'image sont maintenant installés.

ld      a,$07                   ; y inc mode

        call    lcdbusy                 ;

        out     (lcdinstport),a         ;

        ld      a,$80                   ; top row

        ld      d,a                     ;

readl1: call    lcdbusy                 ; set row

        out     (lcdinstport),a         ;

        ld      a,$20                   ; left col

        call    lcdbusy                 ; set col

        out     (lcdinstport),a         ;

        call    lcdbusy                 ;

        in      a,(lcddataport)         ;

        ld      bc,12*256+lcddataport   ; 12 cols

readl2: call    lcdbusy                 ;

        ini                             ; read col

        jr      nz,readl2               ;

        inc     d                       ;

        ld      a,d                     ;

        cp      $BF                     ; bottom row

        jr      nz,readl1               ;

        ld      a,$05                   ; x inc mode

        call    lcdbusy                 ;

        out     (lcdinstport),a         ;

Ceci est magnifique ! Entrée directe à l'écran ! Mon tutorial n'est pas adapté pour couvrir cet algorithme pour lire l'écran entier. Avec un peu de réflexion c'est relativement facile.

exit_interrupt:

        exx                             ;

        ex      af,af'                  ;

        jp      $0038                   ; call standard interrupt handler

Réinverse tous les registres et retourne à l'interruption maître standard.

interrupt_off:

        ld      a,$74                   ; reset apd counter

        ld      ($800A),a               ;

Ne voulons nous pas toujours APD à 0,maintenant c'est fait !

ld      a,$08                   ; turn off lcd

out     (3),a                   ;

ld      a,$01                   ; mask on key interrupts

out     (3),a                   ;

pop     hl                      ; return to when the int occured

exx                             ;

ex      af,af'                  ;

ei                              ; enable interrupts

Eteint la calculatrice. Quand on l'allume…

halt                            ; wait for an interrupt

jp      $0038                   ; call standard interrupt handler

Attendre une interruption et aller à l'interruption maître standard.

delay:  push    af                      ; produce small delay for lcd driver

inc     hl                      ;

dec     hl                      ;

pop     af                      ;

ret                             ;

Délai pour l'écran. Voir le tutorial sur la LCD.

interrupt_end:

;---------------= dialog =--------------

capture:

        .db     "Capture 83",$00

by:

        .db     "By Ian Graf",$00

enabled:

        .db     "Enabled",$00

disabled:

        .db     "Disabled",$00

prog:

        .db     5,"ZCAPTURE"            ; program name for write back

state:

        .db     $00                     ; interrupt state

        .end

Oh mon dieu, c'est fini ! j'espère que vous avez compris. Sinon relisez le tutorial sur le LCD puis de nouveau celui ci, peut être que tout ceci prendra un sens.

Capture.inc

Voici le fichier inclus pour être sur qu'il n'y ait pas de confusion.

#define keydown         $8401

#define lcdinstport     $10

#define lcddataport     $11

#define pictobj         $07

#define tvarpict        $60

#define tpic1           $00

#define op1             $8039

#define usermem         $9327

#define symtable        $FE6E

#define zeroop1         $428E

#define chkfindsym      $442A

#define createpict      $4482

#define puts            $470D

#define newline         $473D

Conclusion

Les programmes d'interruptions sont TRES utiles, mais peuvent être dangereux et ils sont un peu limités du fait que vous devez préservez les registres, vous ne devez jamais appeler une commande. J'ai parlé à un gars aujourd'hui et i m'a dit qu'en changeant la page de la ROM c'était possible… mais laissons ça de côté pour le moment.

Merci à Ian Graf

